

G02CEF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G02CEF takes selected elements from two vectors (typically vectors of means and standard deviations) to form two smaller vectors, and selected rows and columns from two matrices (typically either matrices of sums of squares and cross-products of deviations from means and Pearson product-moment correlation coefficients, or matrices of sums of squares and cross-products about zero and correlation-like coefficients) to form two smaller matrices, allowing re-ordering of elements in the process.

2 Specification

```

SUBROUTINE G02CEF(N, XBAR, STD, SSP, ISSP, R, IR, M, KORDER,
1          XBAR2, STD2, SSP2, ISSP2, R2, IR2, IFAIL)
  INTEGER  N, ISSP, IR, M, KORDER(M), ISSP2, IR2, IFAIL
  real    XBAR(N), STD(N), SSP(ISSP,N), R(IR,N), XBAR2(M),
1          STD2(M), SSP2(ISSP2,M), R2(IR2,M)

```

3 Description

Input to the routine consists of:

- (a) A vector of means:

$$(\bar{x}_1, \bar{x}_2, \bar{x}_3, \dots, \bar{x}_n),$$

where n is the number of input variables.

- (b) A vector of standard deviations:

$$(s_1, s_2, s_3, \dots, s_n)$$

- (c) A matrix of sums of squares and cross-products of deviations from means:

$$\begin{pmatrix} S_{11} & S_{12} & S_{13} & \cdot & \cdot & \cdot & S_{1n} \\ S_{21} & S_{22} & & & & & S_{2n} \\ S_{31} & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ S_{n1} & S_{n2} & \cdot & \cdot & \cdot & \cdot & S_{nn} \end{pmatrix}$$

- (d) A matrix of correlation coefficients:

$$\begin{pmatrix} R_{11} & R_{12} & R_{13} & \cdot & \cdot & \cdot & R_{1n} \\ R_{21} & R_{22} & & & & & R_{2n} \\ R_{31} & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ R_{n1} & R_{n2} & \cdot & \cdot & \cdot & \cdot & R_{nn} \end{pmatrix}$$

- (e) The number of variables, m , in the required subset, and their row/column numbers in the input data, $i_1, i_2, i_3, \dots, i_m$,

$$i \leq i_k \leq n \text{ for } k = 1, 2, \dots, m \text{ (} n \geq 2, m \geq 1 \text{ and } m \leq n).$$

New vectors and matrices are output containing the following information:

- (i) A vector of means:

$$(\bar{x}_{i_1}, \bar{x}_{i_2}, \bar{x}_{i_3}, \dots, \bar{x}_{i_m})$$

(ii) A vector of standard deviations:

$$(s_{i_1}, s_{i_2}, s_{i_3}, \dots, s_{i_m})$$

(iii) A matrix of sums of squares and cross-products of deviations from means:

$$\begin{pmatrix} S_{i_1 i_1} & S_{i_1 i_2} & S_{i_1 i_3} & \cdot & \cdot & \cdot & S_{i_1 i_m} \\ S_{i_2 i_1} & S_{i_2 i_2} & & & & & \cdot \\ S_{i_3 i_1} & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ S_{i_m i_1} & S_{i_m i_2} & \cdot & \cdot & \cdot & \cdot & S_{i_m i_m} \end{pmatrix}$$

(iv) A matrix of correlation coefficients:

$$\begin{pmatrix} R_{i_1 i_1} & R_{i_1 i_2} & R_{i_1 i_3} & \cdot & \cdot & \cdot & R_{i_1 i_m} \\ R_{i_2 i_1} & R_{i_2 i_2} & & & & & \cdot \\ R_{i_3 i_1} & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ \cdot & & & & & & \cdot \\ R_{i_m i_1} & R_{i_m i_2} & \cdot & \cdot & \cdot & \cdot & R_{i_m i_m} \end{pmatrix}$$

Note. For sums of squares of cross-products of deviations about zero and correlation-like coefficients S_{ij} and R_{ij} should be replaced by \tilde{S}_{ij} and \tilde{R}_{ij} in the description of the input and output above.

4 References

None.

5 Parameters

- 1: N — INTEGER *Input*
On entry: the number of variables n , in the input data.
Constraint: $N \geq 2$.
- 2: XBAR(N) — *real* array *Input*
On entry: XBAR(i) must be set to \bar{x}_i , the mean of variable i , for $i = 1, 2, \dots, n$.
- 3: STD(N) — *real* array *Input*
On entry: STD(i) must be set to s_i , the standard deviation of variable i , for $i = 1, 2, \dots, n$.
- 4: SSP(ISSP,N) — *real* array *Input*
On entry: SSP(i, j) must be set to the sum of cross-products of deviations from means S_{ij} (or about zero \tilde{S}_{ij}) for variables i and j , for $i, j = 1, 2, \dots, n$.
- 5: ISSP — INTEGER *Input*
On entry: the first dimension of the array SSP as declared in the (sub)program from which G02CEF is called.
Constraint: ISSP \geq N.
- 6: R(IR,N) — *real* array *Input*
On entry: R(i, j) must be set to the Pearson product-moment correlation coefficient R_{ij} (or the correlation-like coefficient \tilde{R}_{ij}) for variables i and j , for $i, j = 1, 2, \dots, n$.

- 7:** IR — INTEGER *Input*
On entry: the first dimension of the array R as declared in the (sub)program from which G02CEF is called.
Constraint: $IR \geq N$.
- 8:** M — INTEGER *Input*
On entry: the number of variables m , required in the reduced vectors and matrices.
Constraint: $1 \leq M \leq N$.
- 9:** KORDER(M) — INTEGER array *Input*
On entry: KORDER(i) must be set to the number of the original variable which is to be the i th variable in the output vectors and matrices, for $i = 1, 2, \dots, m$.
Constraint: $1 \leq \text{KORDER}(i) \leq N$, for $i = 1, 2, \dots, m$.
- 10:** XBAR2(M) — *real* array *Output*
On exit: the mean of variable i , XBAR(i), where $i = \text{KORDER}(k)$, for $k = 1, 2, \dots, m$. (The array XBAR2 must differ from XBAR and STD).
- 11:** STD2(M) — *real* array *Output*
On exit: the standard deviation of variable i , STD(i), where $i = \text{KORDER}(k)$, for $k = 1, 2, \dots, m$. (The array STD2 must differ from both XBAR and STD).
- 12:** SSP2(ISSP2,M) — *real* array *Output*
On exit: SSP2(k, l) contains the value of SSP(i, j), where $i = \text{KORDER}(k)$ and $j = \text{KORDER}(l)$, for $k, l = 1, 2, \dots, m$. (The array SSP2 must differ from both SSP and R.)
 That is to say: on exit, SSP2(k, l) contains the sum of cross-products of deviations from means S_{ij} (or about zero \tilde{S}_{ij}).
- 13:** ISSP2 — INTEGER *Output*
On entry: the first dimension of the array SSP2 as declared in the (sub)program from which G02CEF is called.
Constraint: ISSP2 \geq M.
- 14:** R2(IR2,M) — *real* array *Output*
On exit: R2(k, l) contains the value of R(i, j), where $i = \text{KORDER}(k)$ and $j = \text{KORDER}(l)$, for $k, l = 1, 2, \dots, m$. (The array R2 must differ from both SSP and R.)
 That is to say: on exit, R2(k, l) contains the Pearson product-moment coefficient R_{ij} (or the correlation-like coefficient \tilde{R}_{ij}).
- 15:** IR2 — INTEGER *Input*
On entry: the first dimension of the array R2 as declared in the (sub)program from which G02CEF is called.
Constraint: IR2 \geq M.
- 16:** IFAIL — INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, $N < 2$,
or $M < 1$.

IFAIL = 2

On entry, $N < M$.

IFAIL = 3

On entry, $ISSP < N$,
or $IR < N$,
or $ISSP < M$,
or $IR2 < M$.

IFAIL = 4

On entry, $KORDER(i) < 1$,
or $KORDER(i) > N$ for some $i = 1, 2, \dots, m$.

7 Accuracy

Not applicable.

8 Further Comments

The time taken by the routine depends on n and m .

The routine is intended primarily for use when a subset of variables from a larger set of variables are to be used in a regression, and is described accordingly. There is however no reason why the routine should not also be used to select specific rows and columns from vectors and arrays which contain any other non-statistical information; the matrices need not be symmetric.

The routine may be used either with sums of squares and cross-products of deviations from means and Pearson product-moment correlation coefficients in connection with a regression involving a constant, or with sums of squares and cross-products about zero and correlation-like coefficients in connection with a regression with no constant.

9 Example

The following program reads in the means, standard deviations, sums of squares and cross-products, and correlation coefficients for four variables. New vectors and matrices are created containing the means, standard deviations, sums of squares and cross-products, and correlation coefficients for the fourth, first and second variables (in that order). Finally these new vectors and matrices are printed.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G02CEF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
INTEGER          N, N2, ISSP, ICORR, ISSPX, ICORRX
PARAMETER       (N=4,N2=3,ISSP=N,ICORR=N,ISSPX=N2,ICORRX=N2)
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
INTEGER          I, IFAIL, J
*      .. Local Arrays ..
real           CORR(ICORR,N), CORRX(ICORRX,N2), SSP(ISSP,N),
+              SSPX(ISSPX,N2), STD(N), STDX(N2), XM(N), XMX(N2)
INTEGER          IORDER(N2)
*      .. External Subroutines ..
EXTERNAL        G02CEF
*      .. Executable Statements ..
WRITE (NOUT,*) 'G02CEF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) (XM(I),I=1,N), (STD(I),I=1,N),
+ ((SSP(I,J),J=1,N),I=1,N), ((CORR(I,J),J=1,N),I=1,N)
WRITE (NOUT,*)
WRITE (NOUT,99999) 'Original vector XM   : ', (XM(I),I=1,N)
WRITE (NOUT,*)
WRITE (NOUT,99999) 'Original vector STD  : ', (STD(I),I=1,N)
WRITE (NOUT,*)
WRITE (NOUT,*) 'Original matrix SSP   :'
WRITE (NOUT,99998) ((SSP(I,J),J=1,N),I=1,N)
WRITE (NOUT,*)
WRITE (NOUT,*) 'Original matrix CORR  :'
WRITE (NOUT,99998) ((CORR(I,J),J=1,N),I=1,N)
WRITE (NOUT,*)
IORDER(1) = 4
IORDER(2) = 1
IORDER(3) = 2
IFAIL = 1
*
CALL G02CEF(N,XM,STD,SSP,ISSP,CORR,ICORR,N2,IORDER,XMX,STDX,SSPX,
+          ISSPX,CORRX,ICORRX,IFAIL)
*
IF (IFAIL.NE.0) THEN
  WRITE (NOUT,99997) 'Routine fails, IFAIL =', IFAIL
ELSE
  WRITE (NOUT,99996) 'New vector XMX   : ', (XMX(I),I=1,N2)
  WRITE (NOUT,*)
  WRITE (NOUT,99996) 'New vector STDX  : ', (STDX(I),I=1,N2)
  WRITE (NOUT,*)
  WRITE (NOUT,*) 'New matrix SSPX   :'
  WRITE (NOUT,99995) ((SSPX(I,J),J=1,N2),I=1,N2)
  WRITE (NOUT,*)
  WRITE (NOUT,*) 'New matrix CORRX  :'
  WRITE (NOUT,99995) ((CORRX(I,J),J=1,N2),I=1,N2)
END IF
STOP

```

```

*
99999 FORMAT (1X,A,4F10.4)
99998 FORMAT (1X,4F10.4)
99997 FORMAT (1X,A,I2)
99996 FORMAT (1X,A,3F10.4)
99995 FORMAT (1X,3F10.4)
      END

```

9.2 Program Data

G02CEF Example Program Data

```

5.8000    2.8000    1.8000    5.4000
5.0695    1.9240    2.5884    4.9800
102.8000  -29.2000  -14.2000  -57.6000
-29.2000  14.8000   -6.2000   6.4000
-14.2000  -6.2000  28.6000  42.4000
-57.6000  6.4000  42.4000  99.2000
1.0000    -0.7486   -0.2619   -0.5704
-0.7486    1.0000   -0.3014    0.1670
-0.2619   -0.3014    1.0000    0.7960
-0.5704    0.1670    0.7960    1.0000

```

9.3 Program Results

G02CEF Example Program Results

Original vector XM : 5.8000 2.8000 1.8000 5.4000

Original vector STD : 5.0695 1.9240 2.5884 4.9800

Original matrix SSP :

```

102.8000  -29.2000  -14.2000  -57.6000
-29.2000  14.8000   -6.2000   6.4000
-14.2000  -6.2000  28.6000  42.4000
-57.6000  6.4000  42.4000  99.2000

```

Original matrix CORR :

```

1.0000   -0.7486   -0.2619   -0.5704
-0.7486    1.0000   -0.3014    0.1670
-0.2619   -0.3014    1.0000    0.7960
-0.5704    0.1670    0.7960    1.0000

```

New vector XMx : 5.4000 5.8000 2.8000

New vector STDx : 4.9800 5.0695 1.9240

New matrix SSPx :

```

99.2000  -57.6000   6.4000
-57.6000 102.8000  -29.2000
 6.4000  -29.2000  14.8000

```

New matrix CORRx :

```

1.0000   -0.5704    0.1670
-0.5704    1.0000   -0.7486
 0.1670   -0.7486    1.0000

```